NBSIR 86-3055



# COUNTING METHODS FOR FATIGUE ANALYSIS WITH RANDOM LOAD HISTORIES: A FORTRAN USER'S GUIDE



Yi-Wen Cheng
Jerry J. Broz

National Bureau of Standards
U.S. Department of Commerce
Boulder, Colorado 80303

August 1986

NBSIR 86-3055



# CYCLE-COUNTING METHODS FOR FATIGUE ANALYSIS WITH RANDOM LOAD HISTORIES: A FORTRAN USER'S GUIDE

Yi-Wen Cheng
Jerry J. Broz

Fracture and Deformation Division
Institute for Materials Science and Engineering
National Bureau of Standards
Boulder, Colorado 80303

August 1986

Sponsored by
U.S. Department of Interior
Minerals Management Service
Reston, Virginia 22091

U.S. DEPARTMENT OF COMMERCE, Malcolm Baldrige, Secretary

NATIONAL BUREAU OF STANDARDS, Ernest Ambler, Director

CONTENTS

LIST OF FIGURES

# CYCLE-COUNTING METHODS FOR FATIGUE ANALYSIS WITH RANDOM LOAD HISTORIES: A FORTRAN USER'S GUIDE

Yi-Wen Cheng
Jerry J. Broz

Fracture and Deformation Division
National Bureau of Standards
Boulder, Colorado 80303

Rainflow and mean crossing-range methods are used in counting the stress ranges and cycles of a random load history. Each method is defined and then applied to a simple random load history example. Fortran IV computer programs were written to make analysis of long random load histories possible. The stress ranges and cycles obtained by these programs have been used for fatigue crack growth analysis under sea-wave loading.

Key words: cycle-counting methods; fatigue of materials; mean crossing-range counting method; rainflow counting method; variable amplitude loading

# INTRODUCTION

Real-life structures, such as airplanes, automobiles, ships, pressure vessels, bridges, offshore structures, etc. are often subjected to cyclical loads that result in structural failure due to fatigue. To avoid any potential fatigue failure, the fatigue life of the structure must be known. The loads that cause structural failure by fatigue are usually complex and random in form. Current approaches to random load fatigue analysis utilize experimentally obtained fatigue data, which are often acquired through constant-amplitude testing methods, such as stress-life (S-N), strain-life ($\varepsilon$-N), or fatigue crack growth rate, to predict the fatigue lives of actual structures subject to random loading. All these commonly used approaches require that stress ranges and cycles of the random load history be defined in order to perform fatigue-life calculation.

Except for constant-amplitude and narrow-band random loadings, where the precise definition of a cycle is clear, the determination of the stress ranges and cycles is a problem. Thus, before any fatigue analysis can be performed, some sort of cycle-counting method to reduce the random load history to proper stress ranges and cycles must be devised. Once a cycle-counting method is established, and the stress ranges and cycles are defined, evaluation of the fatigue damage under random loading is possible. Two of the many cycle-counting methods currently available are the rainflow counting method [1], and the mean crossing-range technique [2]. This guide presents Fortran IV computer programs which utilize the rainflow counting method and the mean crossing-range technique in counting the stress ranges and cycles of random load histories.

## CYCLE-COUNTING METHODS

The purpose of all cycle-counting methods is to reduce a complex random load history so that the results can be compared with S-N and ε-N curves obtained from constant-amplitude testing for predicting the fatigue life of an actual structure subject to random loads. To facilitate the life prediction, the number of cycles and the magnitude of stress ranges in the random load history must be determined. Unlike constant-amplitude and narrow-band load histories, where a cycle and stress ranges have exact definitions, the random load history has no set definition for a cycle or a stress range. It is expected that each cycle-counting method will have a different way of defining a cycle and stress range, resulting in a number of different ways in which the random load history can be counted. With so many variations for counting cycles and stress ranges in a random load history, it is unfortunate that no one method can be satisfactorily proven to be the best. Each method has its advantages as well as its disadvantages. In this guide, two methods for counting random load histories will be presented. These are (1) the rainflow-counting method, and (2) the mean crossing-range technique.

### Rainflow Cycle-Counting Method

Of all the methods currently available, the rainflow cycle-counting technique is the most popular. The rainflow technique was first presented in Japanese in 1968 by Tatsuo Endo, and an English translation was published in 1974 [1]. The method, like other counting methods, defines cycles, stress ranges, and the mean of a random load history. Some of the reasons for its popularity stem from the simplicity of its algorithm and its compatibility with corresponding stress-strain relation when it is applied to a strain history [3]. Another, more important, reason is that the method determines the energy used during the complex loading through the identification of hysterisis loops corresponding to counted stress ranges [4]. Originally, the method was developed to analyze fatigue crack initiation under a random load history. However, using a Fortran IV computer program, the rainflow technique has been extended to analyze fatigue crack propagation in a specimen subject to random load history [5].

The rainflow method gets its name from a metaphorical flow of raindrops down many overlapping "pagoda" roofs, where the peaks and valleys of a random load history are represented by the edge of each roof [1]. In figure 1, a short random load history is turned 90° clockwise. Using some imagination, it can be seen that the load history resembles a series of roofs down which the raindrops must flow. The raindrops can flow in either direction down the roofs (left or right), but they must cover the top of every roof. There are, however, a few conditions which govern the way the raindrops can actually flow. The analysis of the random load history using these conditions is best demonstrated through example.

The rain starts to flow from the first peak, the highest roof, until it reaches the edge, drops off, and lands on another roof. This sequence of events continues until one of the following two conditions is satisfied:

1. The rain falling from a roof above cannot cross the path of rain flowing down a roof. If this occurs, the range is counted and the peak and valley which make up the range are discarded, since they will have no




## Results

| Range | Total Cycles |
|---|---|
| H/I | 1 |
| A/B | 1 |
| D/F | 1 |
| C/G | 1 |

Figure 1. Illustration of rainflow cycle-counting method.

effect on future events. Ranges H/I, A/B, and D/F in figure 1 are good examples of this rule.

2. For a drop falling from the tip of a roof, the flow stops if the falling drop passes opposite either a peak which is more positive than that at the start of the path under consideration or a valley which is more negative than that at the start of the path under consideration [6]. Again, this leads to a range being counted and the points which make up the range being discarded. In figure 1, this can be illustrated using the flow starting at H. The flow stops after I because A is a more negative valley than H.

In this process, if the range counted contains the starting point of the counting sequence, this range comprises only one-half cycle; otherwise, one cycle is counted. In the former case, only the starting point is discarded and the second point becomes the starting point.

If a typical segment of a random load history is repeatedly applied, as is commonly done in laboratory tests, the rainflow count is identical for each subsequent repetition of the history once either the maximum peak or minimum valley is reached for the first time. Detailed procedures for obtaining such a repeating random load history cycle count are as follows:

1. Find the maximum peak (or valley) and begin the rainflow count at that point.
2. Read the next valley (or peak). If out of data, go to step 7.

3. If there are fewer than three points, go to step 2. Form ranges Y (the first two points) and X (the second and third points) using the three most recent peaks and valleys that have not been discarded.
4. Compare ranges X and Y:
   - if $X < Y$, go to step 2.
   - if $X \geqq Y$, go to step 5.
5. Count range Y, discard the peak and valley associated with range Y.
6. Go to step 3.
7. Print the results.

Take the random load history in figure 1 as an example:

1. The maximum peak is found at point G.
2. Read point H.
3. There are fewer than three points. Go to step 2.
4. Read point I.
5. Form range G/H as Y, and range H/I as X.
6. Because X is smaller than Y, go to step 2.
7. Read point A.
8. Form range H/I as Y, and range I/A as X.
9. Because X is greater than Y, go to step 5.
10. Count the range H/I as one cycle, and discard points H and I. Points G and A remain.
11. Go to step 3.
12. There are fewer than three points. Go to step 2.
13. Read point B.

14. Form range G/A as Y, and range A/B as X.

15. Because X is smaller than Y, go to step 2.

16. Read point C.

17. Form range A/B as Y, and range B/C as X.

18. Because X is greater than Y, go to step 5.

19. Count the range A/B as one cycle, discard points A and B. Points G and C remain.

20. Go to step 3.

21. There are fewer than three points. Go to step 2.

22. Read point D.

23. Form range G/C as Y, and range C/D as X.

24. Because X is smaller than Y, go to step 2.

25. Read point E.

26. Because E is an intermediate point, discard point E and go to step 2.

27. Read point F.

28. Form range C/D as Y, and range D/F as X.

29. Because X is smaller than Y, go to step 2.

30. Read point G.

31. Form range D/F as Y, and range F/G as X.

32. Because X is greater than Y, go to step 5.

33. Count the range D/F as one cycle, and discard points D and F. Points G, C, and G remain.

34. Go to step 3.

35. Form range G/C as Y, and range C/G as X.

36. Because X equals Y, go to step 5.

37. Count the range G/C as one cycle, and discard points G and C. Point G remains.

38. Go to step 3.

39. There are fewer than three points. Go to step 2.

40. Out of data, go to step 7.

41. Print the results.

A complete Fortran IV computer program for repeating random load histories is presented in Appendix 1 and an example of the program execution is given in Appendix 3. Many comments have been incorporated into the program; it is intended to be self-explanatory. The program was written to operate in an interactive mode. The inputs, except load-time pairs which are stored in mass storage units, are fed into the computer through a terminal keyboard. Because of the wide variety of computers available, the reader may have to make some minor alterations of the program in order to make it compatible with his data files and system.

### Mean Crossing-Range Technique

The mean crossing-range technique is a modified version of the level-crossing counting method. The mean-load level is used as the basis for counting the cycles and stress ranges. Each time the varying load crosses the mean level, a count is made. After three mean crossings, a cycle is counted and a stress range is measured from the maximum peak to the minimum valley among these crossings. These points are subsequently discarded since they will have no effect on future events. This method is illustrated using the short random load history shown in figure 2.




**Results**

| Range | Total Cycles |
|---|---|
| B/G | 1 |
| C/G | 1 |

Figure 2. Illustration of mean croosing-range cycle-counting technique.

Starting at point G, only consecutive positive peaks and negative valleys with a mean crossing in between will be considered for analysis. The points G and H are not read into the calculations because there is no mean crossing between them; also, they are of the same sign. There are, however, mean crossings between points I and A, A and B, and B and C. These points, are in turn used to define the cycle and stress range. As shown in figure 2, there are three mean crossings between the points I and C, thus, the range which spans from B to C is counted as one cycle. Once the cycle and stress range are counted, the points preceding point C are discarded from any further calculations, and the count continues from point C. The counted cycles and stress ranges are also given in figure 2. A complete Fortran IV computer program using the mean crossing-range technique for counting random load histories is presented in Appendix 2 and an example of the program execution is given in Appendix 3. Because of the wide variety of computers available, the reader may have to make some minor alterations of the program in order to make it compatible with his data files and system. Again, many comments have been incorporated into the program; it is intended to be self-explanatory. The program was written to be interactive. The inputs, except load-time pairs which are stored in mass storage units, are fed into the computer through a terminal keyboard.

## SUMMARY

Many methods for counting the cycles and stress ranges in a random load history exist. This guide presents two of these methods, the rainflow counting technique and the mean crossing-range method, in the form of simple examples and complete Fortran IV computer programs. These easy to follow programs have been successfully run on repeating random load histories of different irregularity factors (the number of mean crossings divided by the number of peaks plus valleys) [2,5].

## REFERENCES


1. Endo, T., Mitsunaga, K., Takahashi, K., Kobayashi, K., and Matsuishi, M.,"Damage Evaluation of Metals for Random or Varying Loading - Three Aspects of the Rainflow Method," Proc 1974 Symposium on Mechanical Behavior of Materials, Vol. 1 (The Society of Materials Science, Japan 1974), pp. 371-380.
2. Cheng, Y. W., "The Fatigue Crack Growth of a Ship Steel in Saltwater under Spectrum Loading," Int. J. of Fatigue, Vol. 7, No. 2, 1985, pp. 95-100.
3. Anzia, H., and Endo, T., "On Site Indication of Fatigue Damage under Complex Loading," Int. J. of Fatigue, Vol. 1, No. 1, 1979, pp. 49-57.
4. Downing, S. D., and Socie, D. F., "Simplified Rainflow Counting Algorithms," Int. J. of Fatigue, Vol. 4, No. 1, 1982, pp. 31-40.
5. Cheng, Y. W., "Fatigue Crack Growth Analysis under Seawave Loading," submitted for publication in the International Journal of Fatigue.
6. Gurney, T. R., Fatigue of Welded Structures, second edition, 1979, Cambridge University Press, Cambridge, U. K., pp. 378-380.

## APPENDIX 1

## COMPUTER PROGRAM LIST FOR RAINFLOW CYCLE-COUNTING METHOD

## (FOR REPEATING LOAD HISTORIES)

```
0001          PROGRAM CYCLE1
      C
      C-- PURPOSE:
      C--    EXAMINE A RANDOM LOAD HISTORY, WHICH IS ON
      C--    TWO SEPARATE DISKS, FIND THE MAXIMUM PEAK, COUNT
      C--    THE NUMBER OF POINTS, AND USE THE RAINFLOW
      C--    COUNTING METHOD TO COUNT THE CYCLES AND THE
      C--    STRESS RANGES.
      C-- USAGE:
      C--    THE PROGRAM CYCLE1 IS ON THE SYSTEM DISK, SO THAT
      C--    THE GREATEST AMOUNT OF DATA CAN BE ANALYZED. THUS
      C--    IT MUST BE RUN WITH THE COMMAND 'R CYCLE1'.
      C-- DESCRIPTION OF PARAMETERS:
      C--    NONE.
      C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
      C--    FILES
      C--    MAXMIN
      C--    HSTINT
      C--    TEXPO
      C--    RANFLO(YZ,IJK,YHAT)
      C--    HSTGRM(RANGE)
      C--    YESNO(IYESNO)
      C-- COMMENTS:
      C--    THIS PROGRAM IS COMPOSITE OF SEVERAL SUBROUTINES
      C--    EACH HAVING A SEPARATE PURPOSE.  THESE SUBROUTINES
      C--    ARE COMBINED IN A LIBRARY FILE (RANFIL) WHICH IN
      C--    TURN IS LINKED WITH THE MAIN PROGRAM CYCLE1.
      C--
      C--    THE DATA ANALYZED IS ON TWO SEPERATE DISKS:
      C--        RIGHT-SIDE DISK - ONE FILE OF JUST DATA. THIS
      C--         DISK IS ALWAYS READ FIRST.
      C--        LEFT-SIDE DISK  - SYSTEM DISK WITH CYCLE1
      C--         PROGRAM, AS WELL AS, ANOTHER DATA FILE (VAXT.DAT)
      C--         WHICH IS A CONTINUATION OF THE FILE ON THE RIGHT
      C--         SIDE DISK.
      C-- PROGRAM WRITTEN BY YI-WEN CHENG AND JERRY J. BROZ
      C-- THIS PROGRAM IS A PROPERTY OF THE U.S. GOVERNMENT
      C-- AND NOT SUBJECT TO COPYRIGHT
      C
0002          COMMON/EXPO/YZ,IJK,YHAT
0003          COMMON/FILES/INFILE
0004          COMMON/HSTINT/NINC,XINC,YRANGE(100),XRANGE(100)
0005          COMMON/MAXMIN/APEX,AVALLY,XPOINT
      C
      C-- THE FILE VAXT.DAT ON THE LEFT-SIDE DISK IS ASSIGNED
      C-- THE LABEL FTN60.DAT.
0006          CALL ASSIGN(60,'SY:VAXT.DAT',11,,,)
      C
0007          TYPE 10
0008  10      FORMAT(/////' -- THIS CYCLE1 PROGRAM IS USED FOR',/,
             1 ' THE LEFT AND RIGHT SIDE DISKS.  THE INPUT FILE',/,
             1 ' ON THE LEFT DISK SHOULD BE VAXT.DAT AND THE',/,
             1 ' INPUT FILE ON THE RIGHT DISK NEEDS TO BE DES-',/,
             1 ' IGNATED.  PLEASE MAKE SURE THE CORRECT DISKS',/,
```

```
             1 ' ARE IN THE RESPECTIVE DRIVES BEFORE STARTING.')
       C
       C-- OBTAIN THE NAME OF THE INPUT FILE WHICH IS ON THE
       C-- RIGHT SIDE DISK.
0009          CALL FILES
       C
       C-- READ THE DATA FROM BOTH THE RIGHT AND LEFT DISKS, THEN FIND
       C-- THE MAXIMUM PEAK, THE MINIMUM VALLEY, AND THE TOTAL NUMBER
       C-- OF DATA POINTS PRESENT IN THE COMBINED FILES.
0010          CALL MAXMIN
       C
       C-- DESIGNATE THE INTERVAL TO BE USED FOR PRINTING OUT THE
       C-- HISTOGRAM OF THE STRESS RANGES.
0011          CALL HSTINT
       C
       C-- PICK THE STARTING EXPONENT, THE NUMBER OF CALCULATIONS,
       C-- AND THE INCREMENT OF THE EXPONENT FOR FINDING THE DESIRED
       C-- VALUES OF H-ROOT.
0012          CALL TEXPO
       C
       C-- EXAMINE THE RANDOM LOAD-TIME HISTORY ON THE TWO DISKS,
       C-- DEFINE AND COUNT THE CYCLES AND STRESS RANGES USING THE
       C-- RAINFLOW COUNTING METHOD.
0013          CALL RANFLO(Y2,IJK,YHAT)
0014          STOP
0015          END
```

```
      SUBROUTINE FILES
C
C-- PURPOSE:
C--   DESIGNATE WHICH INPUT FILE WILL BE USED IN THE
C--   CALCULATIONS.  THIS DATA FILE SHOULD BE FOUND
C--   ON THE RIGHT-SIDE DISK.
C-- USAGE:
C--   CALL FILES
C-- DESCRIPTION OF PARAMETERS:
C--   INFILE - THE NUMBER OF THE FILE TO BE EXAMINED
C--   IANS   - YES=1, OR NO=0
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   YESNO(IYESNO)
C-- COMMENTS:
C--   NONE.
C
      COMMON/FILES/INFILE
20    TYPE 30
C
C-- DESIGNATE INPUT FILE TO BE USED
30    FORMAT(/,' TYPE IN THE NO. OF INPUT FILE',/)
      ACCEPT 40, INFILE
40    FORMAT(I3)
      TYPE 50, INFILE
50    FORMAT(//,'-- THERE WILL BE A SLIGHT WAIT',/,
     1 ' WHILE THIS FILE IS READ. --',//,
     1 ' THE INPUT FILE IS',I4,//,
     1' IS THIS INFORMATION CORRECT?')
      CALL YESNO(IANS)
      IF (IANS .EQ. 0) GO TO 20
      RETURN
      END
```

```
      SUBROUTINE HSTGRM(RANGE)
C
C-- PURPOSE:
C--   CALCULATES THE SIZE OF THE INTERVAL FOR THE HISTOGRAM
C--   PLACEMENT OF RANGES, PLACES THE OBTAINED STRESS RANGES
C--   INTO THE APPROPRIATE INTERVAL, AND THEN COUNTS THE
C--   NUMBER OF TIMES THE RANGES FALL WITHIN A GIVEN
C--   INTERVAL. NOTE, THIS IS HISTOGRAM DATA, AND NOT A
C--   HISTOGRAM.
C-- USAGE:
C--   CALL HSTGRM(RANGE)
C-- DESCRIPTION OF PARAMETERS:
C--   XRANGE - ARRAY OF THE ENDPOINTS OF THE INTERVAL
C--            USED FOR THE HISTOGRAM PLACEMENT
C--   RANGE  - STRESS RANGE OBTAINED FROM RAINFLOW
C--   XDEL1  - CHECKS TO SEE IF THE RANGE IS ABOVE THE
C--            LOWER ENDPOINT OF THE HISTOGRAM INTERVAL
C--   XDEL2  - CHECKS TO SEE IF THE RANGE IS BELOW THE
C--            UPPER ENDPOINT OF THE HISTOGRAM INTERVAL
C--   YRANGE - ARRAY TO COUNT THE NUMBER OF TIMES A RANGE
C--            MAGNITUDE FALLS WITHIN A GIVEN INTERVAL
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   NONE.
C-- COMMENTS:
C--   DOUBLE PRECISION IS USED TO MAKE SURE THAT ALL THE
C--   RANGES GET COUNTED AND PLACED INTO THE HISTOGRAM.
C--   MAINLY USED TO INCREASE THE ACCURACY OF THE
C--   COMPARISON/PLACEMENT TEST.
C
      IMPLICIT DOUBLE PRECISION (A,H,R,X)
      COMMON/HSTINT/NINC,XINC,YRANGE(100),XRANGE(100)
C
C-- INITIALIZE THE PARAMETER
      XRANGE(1)=0.
C
C-- INCREMENT STEP TO ESTABLISH INTERVALS
      DO 10, I=2,NINC+1
      XRANGE(I)=XRANGE(I-1)+XINC
C
C-- * COMPARISON/PLACEMENT TEST *
C-- COMPARISON OF THE RANGES FOR HISTOGRAM PLACEMENT
      XDEL1=RANGE-XRANGE(I-1)
      XDEL2=RANGE-XRANGE(I)
      IF (XDEL1 .GT. 1.D-5 .AND. XDEL2 .LE. 1.D-5)
C
C-- COUNTS THE NUMBER OF OCCURANCES WITHIN AN INTERVAL
     1 YRANGE(I-1)=YRANGE(I-1)+1.
10    CONTINUE
      RETURN
      END
```

```
      SUBROUTINE HSTINT
C
C-- PURPOSE:
C--   PROMPTS FOR THE NUMBER OF INERVALS TO BE USED IN
C--   SETTING UP A HISTGRAM FOR THE STRESS RANGES OBTAINED
C--   FROM RAINFLOW.  ONCE THE NUMBER OF INTERVALS IS
C--   KNOWN, THE INCREMENT OF THE HISTOGRAM CAN BE ESTABLISHED.
C-- USAGE:
C--   CALL HSTINT
C-- DESCRIPTION OF PARAMETERS:
C--   NINC   - THE TOTAL NUMBER OF INTERVALS BETWEEN
C--            EXTREMAS FOR THE HISTOGRAM
C--   RNGMAX - THE MAXIMUM POSSIBLE RANGE FOR THE HISTOGRAM
C--   XNINC  - THE REAL NUMBER CONVERSION OF NINC
C--   XINC   - THE INCREMENT FOR THE HISTOGRAM
C--   YRANGE - ARRAY TO COUNT THE NUMBER OF TIMES A RANGE
C--            MAGNITUDE FALLS WITHIN A GIVEN INTERVAL
C--   XRANGE - ARRAY OF THE ENDPOINTS OF THE INTERVALS
C--            USED FOR THE HISTOGRAM PLACEMENT
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   YESNO(IYESNO)
C-- COMMENTS:
C--   DOUBLE PRECISION IS USED TO INCREASE THE ACCURACY
C--   OF THE CALCULATIONS.
C
      IMPLICIT DOUBLE PRECISION (A,H,R,X)
      COMMON/MAXMIN/APEX,AVALLY,XPOINT
      COMMON/HSTINT/NINC,XINC,YRANGE(100),XRANGE(100)
C
C
C-- THE MAXIMUM POSSIBLE RANGE FOR THE HISTOGRAM IS FOUND
      RNGMAX=APEX-AVALLY
20    TYPE 30
30    FORMAT(/,' TYPE IN THE NUMBER OF DESIRED INTERVALS',/,
     1' BETWEEN THE EXTREMA',I4,/)
      ACCEPT 40,NINC
40    FORMAT(I3)
      TYPE 50,NINC
50    FORMAT(/,' THE NUMBER OF INTERVALS IS: ',I5,/,
     1' IS THIS INFORMATION CORRECT?')
      CALL YESNO(IANS)
      IF (IANS .EQ. 0) GO TO 20
C
C-- REAL NUMBER CONVERSION
      XNINC=FLOAT(NINC)
      XINC=RNGMAX/XNINC
C
C-- INITIALIZE YRANGE FOR THE HISTOGRAM COUNT
      DO 60,I=1,NINC
      YRANGE(I)=0.
60    CONTINUE
      RETURN
      END
```

```
      SUBROUTINE MAXMIN
C
C-- PURPOSE:
C--   TO READ DATA FROM BOTH LEFT AND RIGHT DISKS, FIND THE
C--   MAXIMUM PEAK,THE MINIMUM VALLEY, AND THEN COUNT THE TOTAL
C--   NUMBER OF DATA POINTS.
C-- USAGE:
C--   CALL MAXMIN
C-- DESCRIPTION OF PARAMETERS:
C--   IREAD  - DETERMINES WHICH DISK IS TO BE READ
C--         0   - INFILE
C--         <>0 - VAXT.DAT (FTN60.DAT)
C--   XPOINT - THE TOTAL NUMBER OF DATA POINTS
C--   APEX   - MAXIMUM PEAK
C--   AVALLY - MINIMUM VALLEY
C--   IDAY   - NUMBER DAYS OF RANDOM LOAD
C--   IHOUR  - NUMBER HOURS OF RANDOM LOAD
C--   IMIN   - NUMBER OF MINUTES OF RANDOM LOAD
C--   XSEC   - NUMBER OF SECONDS OF RANDOM LOAD
C--   XH     - LOAD CORRESPONDING TO ABOVE TIMES
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   NONE.
C-- COMMENTS:
C--   DOUBLE-PRECISION WAS USED TO INCREASE THE ACCURACY
C--    OF THE CALCULATIONS.
C
      IMPLICIT DOUBLE PRECISION (A,H,R,X)
      COMMON/FILES/INFILE
      COMMON/MAXMIN/APEX,AVALLY,XPOINT
C
C-- INITIALIZE THE PARAMETERS
      IREAD=0
      XPOINT=0.
      APEX=-10000.0
      AVALLY=10000.0
C
C-- XPOINT IS USED AS A COUNTER FOR EACH DATA FILE
C-- AND IS INCREASED BY ONE EACH TIME A DATA POINT IS
C-- READ, RESULTING IN A TOTAL COUNT OF ALL DATA POINTS
C-- IN THE INPUT FILES.
10    XPOINT=XPOINT+1.
C
C-- THE LOAD-TIME HISTORY IS READ FROM EACH DATA FILE.
C-- THE READING STARTS WITH THE RIGHT-SIDE DISK AND ONCE
C-- THAT DISK IS FINISHED, THE LEFT-SIDE DISK IS READ.
      IF (IREAD .NE. 0) GO TO 21
C
C-- ONCE THE END OF THE FILE IS REACHED, IREAD MUST BE MADE
C-- GREATER THAN ZERO TO INDICATE THE LEFT DISK SHOULD BE READ.
      READ(INFILE,20,END=30) IDAY,IHOUR,IMIN,XSEC,XH
20    FORMAT(3I5,2F14.5)
      GO TO 25
C
C-- ONCE THE LEFT-SIDE DISK HAS BEEN COMPLETELY READ,
```

```
C-- THE RESULTS CAN BE PRINTED OUT IF SO DESIRED.
21    READ(60,20,END=40) IDAY,IHOUR,IMIN,XSEC,XH
C
C-- THE COMPARISON TESTS TO FIND THE PEAK AND VALLEY OF FILES
25    IF (XH .GT. APEX) APEX=XH
      IF (XH .LT. AVALLY) AVALLY=XH
      GO TO 10
30    IREAD=IREAD+1
      GO TO 21
C
C-- PRINTS THE PEAK AND VALLEY
40    PRINT 41,APEX,AVALLY
41    FORMAT(//' THE PEAK IS ',D14.5,/,' THE VALLEY IS ',D14.5)
      XPOINT=XPOINT-1.
C
C-- PRINTS THE NUMBER OF DATA POINTS PRESENT IN BOTH FILES
      PRINT 60,XPOINT
60    FORMAT(/' THERE ARE ',D14.5,' POINTS.'//)
      RETURN
      END
```

```
      SUBROUTINE RANFLO(YZ,IJK,YHAT)
C
C-- PURPOSE:
C--   STARTING AT THE MAXIMUM PEAK OF THE INPUT FILES
C--   COUNT THE CYCLES AND STRESS RANGES IN A RANDOM
C--   LOAD-TIME HISTORY USING THE RAINFLOW-COUNTING
C--   METHOD.  THEN MAKE A HISTOGRAM USING THE OBTAINED
C--   STRESS RANGES AND CALCULATE THE VALUE(S) FOR
C--   H-ROOT WITH THE GIVEN EXPONENT(S)
C-- USAGE:
C--   CALL RANFLO(YZ,IJK,YHAT)
C-- DESCRIPTION OF PARAMETERS:
C--   JR     - COUNTER USED WHEN MANIPULATING DATA
C--   IREAD  - DETERMINES WHICH DISK IS TO BE READ
C--                  0    - INFILE
C--                  <>0 - VAXT.DAT (FTN60.DAT)
C--   YPOINT - COUNTER OF THE CURRENT DATA POINTS BEING
C--            READ FROM THE INPUT FILES
C--   XNUMBR - COUNTER FOR THE TOTAL NUMBER STRESS RANGES
C--   X2     - EXPONENT FOR H-ROOT
C--   H      - ARRAY INTO WHICH THE PEAKS AND VALLEYS
C--            ARE READ AND THEN USED IN RAINFLOW COUNTING
C--   HBARR  - ARRAY INTO WHICH THE SUMMATION OF THE STRESS
C--            RANGES RAISED TO THE X2 POWER ARE PLACED
C--   HROOT  - ARRAY INTO WHICH THE CALCULATED VALUE OF
C--            H-ROOT IS PLACED
C--   APEX   - MAXIMUM PEAK OF THE LOAD-TIME HISTORY
C--   XDIFF  - USED TO LOCATE THE PEAK OF THE INPUT FILES
C--            SO THAT RAINFLOW CAN BE STARTED AT THAT POINT
C--   XSIGN1 - SLOPE OF CURRENTLY CONSIDERED RANGE
C--   XSIGN2 - SLOPE OF PREVIUOSLY CONSIDERED RANGE
C--   ABSIGN - USED TO CHECK IF CONSECUTIVE DATA POINTS (PEAK
C--            OR VALLEY) ARE EQUAL
C--   XSIGN  - USED TO CHECK IF XSIGN1 AND SIGN2 ARE OF THE
C--            SAME SIGN (+/-)
C--   X , XY - RANGES FORMED USING THE THREE MOST RECENT
C--            PEAKS AND VALLEYS
C--   XYMNSX - DIFFERENCE OF RANGE X AND RANGE XY, USED TO CHECK
C--            IF A COMPLETE CYCLE WAS FORMED
C--                  0<  - NO CYCLE FORMED, NEXT POINT MUST BE READ
C--                  0>= - CYCLE FORMED AND COUNTED
C--   RANGE  - THE COUNTED STRESS RANGE (XY)
C--   YXZ    - EXPONENT FOR HBARR CACULATION
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   HSTGRM(RANGE)
C-- COMMENTS:
C--   WITH THIS SUBROUTINE, THE RAINFLOW METHOD IS EXTENDED
C--   TO THE ANALYSIS OF FATIGUE CRACK PROPAGATION IN A
C--   SPECIMEN SUBJECT TO A RANDOM LOAD-TIME HISTORY.
C
      IMPLICIT DOUBLE PRECISION (A,H,R,X)
      COMMON/FILES/INFILE
      COMMON/MAXMIN/APEX,AVALLY,XPOINT
      COMMON/HSTINT/NINC,XINC,YRANGE(100),XRANGE(100)
```

```
      DIMENSION H(100),HROOT(100),HBARR(100)
      REWIND INFILE
      REWIND 60
C
C-- INTITAILIZE THE PARAMETERS
      YXZ=YZ
      IREAD=0
      YPOINT=1.
      H(1)=APEX
      JR=1
      XNUMBR=0.
      DO 15,I=1,IJK
      HBARR(I)=0.
      HROOT(I)=0.
15    CONTINUE
C
C-- WITH THE APEX VALUE (MAXIMUM PEAK) PREVIOUSLY
C--  OBTAINED, THE INPUT FILES ARE RE-EXAMINED TO FIND
C--  THE LOCATION OF THE APEX SO THAT THE RAINFLOW
C--  COUNTING METHOD CAN BE STARTED AT THE MAXIMUM PEAK.
C--  BOTH OF THE INPUT FILES MUST BE EXAMINED.
      IF (IREAD .NE. 0) GO TO 25
10    READ(INFILE,20,END=27)IDAY,IHOUR,IMIN,XSEC,XH
20    FORMAT(3I5,2D14.5)
      GO TO 26
25    READ(60,20,END=999)IDAY,IHOUR,IMIN,XSEC,XH
26    XDIFF=DABS(XH-APEX)
      IF (XDIFF .GT. 1.D-5) GO TO 10
      GO TO 30
27    IREAD=IREAD+1
      GO TO 25
30    JR=JR+1
      YPOINT=YPOINT+1.
C
C-- READS IN THE PEAKS AND VALLEYS USED FOR RAINFLOW COUNTING
      IF (IREAD .GE. 1) GO TO 35
      READ(INFILE,20,END=35)IDAY,IHOUR,IMIN,XSEC,H(JR)
      GO TO 40
35    READ(60,20,END=100)IDAY,IHOUR,IMIN,XSEC,H(JR)
38    IREAD=IREAD+1
C
C-- ONLY FOUR DATA POINTS ARE ALLOWED
C--  IN THE H ARRAY AT ONE TIME.
40    IF (JR .LT. 4) GO TO 30
C
C-- CALCULATES THE SLOPES OF THE CURVE BEING EXAMINED
      XSIGN1=H(JR)-H(JR-1)
      ABSIGN=DABS(XSIGN1)
      XSIGN2=H(JR-1)-H(JR-2)
C
C-- CHECKS IF SUCCESSIVE POINTS ARE EQUAL
C--  IF IT SO HAPPENS, ONLY ONE OF THE POINTS IS CONSIDERED
C--  FOR FUTURE CALCULATIONS.
      IF (ABSIGN .GT. 1.D-5) GO TO 50
```

```
      JR=JR-1
      GO TO 30
C
C-- CHECKS IF TWO CONSECUTIVE SLOPES ARE OF THE SAME SIGN (+/-)
C--  THIS IS REALLY TO MAKE SURE THAT ONLY PEAKS AND VALLEYS ARE
C--  USED IN THE RANGE CALCULATIONS.  THUS, ONLY THE PEAK AND
C--  THE VALLEY IS READ, THE INTERMEDIATE POINTS ARE DISCARDED.
50    XSIGN=(XSIGN1/XSIGN2)
      IF (XSIGN .LT. 0.) GO TO 60
      H(JR-1)=H(JR)
      JR=JR-1
      GO TO 30
C
C-- THE RANGES ARE FORMED, AND USING THE RULES OF RAINFLOW,
C--  ARE COMPARED.  IF THE CONDITIONS OF RAINFLOW ARE MET,
C--  THE STRESS RANGE IS COUNTED.
60    X=DABS(H(JR-1)-H(JR-2))
      XY=DABS(H(JR-2)-H(JR-3))
      XYMNSX=XY-X
      IF (XYMNSX .GT. 1.D-5) GO TO 30
      RANGE=XY
C
C-- THE STRESS RANGES ARE PUT INTO A HISTOGRAM
      CALL HSTGRM(RANGE)
C
C-- THE VALUES OF H-BARR ARE CALCULATED USING THE STRESS
C--  RANGES JUST OBTAINED BY RAINFLOW.
      DO 70,I=1,IJK
      HBARR(I)=(RANGE**YXZ)+HBARR(I)
      YXZ=YXZ+YHAT
      IF (I .EQ. IJK) YXZ=YZ
70    CONTINUE
      XNUMBR=XNUMBR+1.
C
C-- THE PEAK AND THE VALLEY OF THE COUNTED RANGE ARE
C--  DISCARDED SINCE THEY HAVE NO BEARING ON FUTURE EVENTS.
      JR=JR-2
      H(JR)=H(JR+2)
      H(JR-1)=H(JR+1)
C
C-- CHECK IF ALL THE PEAKS AND VALLEYS HAVE BEEN EXAMINED
C--  BY THE RAINFLOW COUNTING METHOD.
      IF ((JR-1) .EQ. 1 .AND. YPOINT .GT. XPOINT) GO TO 200
      GO TO 40
100   IREAD=0
      REWIND INFILE
      REWIND 60
      JR=JR-1
      YPOINT=YPOINT-1.
      GO TO 30
C
C-- THE RESULTS OF THE HISTOGRAM ARE PRINTED OUT.
200   PRINT 201
201   FORMAT('   ----------- RANGE -----------',10X,
```

```
     1 '    -- CYCLE --')
       DO 240, I=1,NINC
 230   PRINT 231,XRANGE(I),XRANGE(I+1),YRANGE(I)
 231   FORMAT(D14.5,'  ---',D13.5,10X,E14.5)
 240   CONTINUE
 245   PRINT 246,XNUMBR
 246   FORMAT(/' THE NUMBER OF RANGES IS: ',D14.5)
C
C-- THE H-ROOT OF THE RANGES IS NOW CALCULATED USING
C--  THE RESPECTIVE EXPONENT, AND THE RESULTS ARE PRINTED OUT.
       DO 255,I=1,IJK
       HROOT(I)=(HBARR(I)/XNUMBR)**(1./Y2)
 250   PRINT 251,Y2,HROOT(I)
 251   FORMAT(//' WITH THE EXPONENT OF: ',E14.5,/,
     1 ' THE H-ROOT OF THE RANGES IS: ',D14.5)
       Y2=Y2+YHAT
 255   CONTINUE
       GO TO 999
 990   TYPE 991
 991   FORMAT(' SOMETHING IS WRONG!!!!!')
 999   RETURN
       END
```

```
      SUBROUTINE TEXPO
C
C-- PURPOSE:
C--   DESIGNATES THE STARTING EXPONENT, THE NUMBER OF H-ROOT
C--   CALCULATIONS, AND THE INCREMENT BY WHICH THE EXPONENT
C--   IS INCREASED FOR EACH H-ROOT CALCULATION.
C-- USAGE:
C--   CALL TEXPO
C-- DESCRIPTION OF PARAMETERS:
C--   YZ      - STARTING EXPONENT FOR THE H-ROOT CALCULATIONS
C--   IJK     - THE NUMBER OF TIMES H-ROOT IS TO BE CALCULATED
C--   YHAT    - THE INCREMENT BY WHICH THE EXPONENT IS INCREASED
C--             FOR EACH H-ROOT CALCULATION
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   YESNO(IYESNO)
C-- COMMENTS:
C--   NONE.
C
      COMMON/EXPO/YZ,IJK,YHAT
   20 TYPE 30
   30 FORMAT(/,' TYPE STARTING EXPONENT FOR HROOT')
      ACCEPT 40,YZ
   40 FORMAT(F7.4)
      TYPE 42
   42 FORMAT(/,' TYPE THE NUMBER OF H-ROOT CALCULATIONS')
      ACCEPT 45,IJK
   45 FORMAT(I5)
      TYPE 46
   46 FORMAT(/,' TYPE THE INCREMENT FOR H-ROOT EXPONENT')
      ACCEPT 47,YHAT
   47 FORMAT(F7.5)
      TYPE 50,YZ
   50 FORMAT(/,' THE STARTING EXPONENT OF HROOT IS: ',F7.4,/)
      TYPE 55,IJK
   55 FORMAT(' THE NUMBER OF H-ROOT CALCULATIONS IS: ',I5,/)
      TYPE 60,YHAT
   60 FORMAT(' THE INCREMENT OF H-ROOT EXPONENT IS: ',F7.5,//,
     1 ' IS THIS INFORMATION CORRECT?')
      CALL YESNO(IANS)
      IF (IANS .EQ. 0) GO TO 20
      RETURN
      END
```

```
      SUBROUTINE YESNO(IYESNO)
C
C-- THIS IS A SIMPLE DOUBLE CHECK TEST PROGRAM
C-- CAN BE USED ANYWHERE AN INPUT IS TO BE
C-- DOUBLE CHECKED.
      LOGICAL*1 ANS(20)
      LOGICAL*1 CHARN
      DATA CHARN/1HN/
      TYPE 2
2     FORMAT(' ANSWER YES OR NO')
      ACCEPT 3, ANS
3     FORMAT(20A1)
      DO 70 I=1,20
70    IF(ANS(I).EQ.CHARN)GOTO  80
      IYESNO=1
      RETURN
80    IYESNO=0
      END
```

## APPENDIX 2

## COMPUTER PROGRAM LIST FOR MEAN CROSSING-RANGE TECHNIQUE

```
0001        PROGRAM HBARNO
      C
      C-- PURPOSE:
      C--    EXAMINE A RANDOM LOAD HISTORY, WHICH IS ON TWO
      C--    SEPERATE DISKS, FIND THE MAXIMUM PEAK, AND THE MINIMUM
      C--    VALLEY, COUNT THE NUMBER OF DATA POINTS, AND USE THE
      C--    ZERO CROSSING MEAN METHOD TO COUNT THE CYCLES AND STRESS
      C--    RANGES PRESENT.
      C--USAGE:
      C--    THE PROGRAM HBARNO IS ON THE SYSTEM DISK, SO THAT THE
      C--    GREATEST AMOUNT OF DATA CAN BE ANALYZED.  THUS, IT MUST
      C--    BE RUN WITH THE COMMAND 'R HBARNO'
      C-- DESCRIPTION OF PARAMETERS:
      C--    IDAY    - NUMBER OF DAYS OF RANDOM LOAD
      C--    IHOUR   - NUMBER OF HOURS OF RANDOM LOAD
      C--    IMIN    - NUMBER OF MINUTES OF RANDOM LOAD
      C--    SEC     - NUMBER OF SECONDS OF RANDOM LOAD
      C--    XT      - LOAD CORRESPONDING TO ABOVE TIMES
      C--    YX2,X2  - EXPONENT OF HBARN
      C--    IREAD   - DETERMINES WHICH DISKIS TO BE READ
      C--                  0     - INFILE
      C--                  0<>  - VAXT.DAT (FTN60.DAT)
      C--    CYCLE   - COUNTS THE NUMBER OF CYCLES FOUND BY USING
      C--            THE ZERO CROSSING MEAN METHOD
      C--    N       - COUNTER USED WHEN MANIPULATING DATA
      C--    ISIGN   - ARRAY USED FOR COMPARING DATA POINTS WHEN
      C--            CYCLE COUNTING
      C--                  XT<0   - ISIGN=0
      C--                  XT=>0  - ISIGN=1
      C--    XTMAX   - ARRAY USED IN CALCULATING THE MAGNITUDE OF
      C--            THE STRESS RANGES
      C--    HBAR    - ARRAY INTO WHICH THE SUMMATION OF THE STRESS
      C--            RANGES RAISED TO THE X2 POWER ARE PLACED
      C--    HBARN   - ARRAY INTO WHICH THE CALCULATED VALUE OF
      C--            HBARN (H-ROOT) IS PLACED
      C--    XTABS   - THE ABSOLUTE VALUE OF THE PEAK OR VALLEY
      C--    RANGE   - THE MAGNITUDE OF THE COUNTED STRESS RANGE
      C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
      C--    TEXPO
      C--    INPUT
      C--    MXMN
      C--    HSTNT2
      C--    HISTGM(RANGE)
      C-- COMMENTS:
      C--    THE CALCULATION TO COMPUTE THE VALUE OF HBARN IS DONE
      C--    USING THE EXPONENT IN THE PARIS EQUATION.  THE ONLY
      C--    RANGES COUNTED ARE THOSE WITH THREE ZERO CROSSINGS.
      C-- PROGRAM WRITTEN BY YI-WEN CHENG AND JERRY J. BROZ
      C-- THIS PROGRAM IS A PROPERTY OF THE U.S. GOVERNMENT
      C-- AND NOT SUBJECT TO COPYRIGHT
      C
0002        COMMON/EXPO/YZ,IJK,YHAT
0003        COMMON/MXMN/APEX,AVALLY,XPOINT
0004        COMMON/HSTNT2/NINC,XINC,YRANGE(100),XRANGE(100)
```

```
      COMMON/INPUT/INFILE
      DIMENSION ISIGN(3),XTMAX(3),HBAR(100),HBARN(100)
C
C-- THE FILE VAXT.DAT ON THE LEFT-SIDE DISK IS ASSIGNED THE
C-- LABEL FTN60.DAT.
      CALL ASSIGN(60,'SY:VAXT.DAT',11,,,)
      TYPE 99
99    FORMAT(/////,' THIS PROGRAM COUNTS THE CYCLES AND CALCULATES',/,
     1 ' THE VALUE OF H-ROOT USING THE HBARN COUNTING METHOD.')
C
C-- PICK THE STARTING EXPONENT, THE NUMBER OF HBARN CALCULATIONS,
C-- AND THE INCREMENT OF THE EXPONENT FOR FINDING THE DESIRED
C-- VALUES OF HBARN (H-ROOT).
      CALL TEXPO
C
C-- INITIALIZE THE PARAMETERS
      XNUMBR=0.
      YXZ=YZ
      IREAD=0
      CYCLE=0.
      N=1
      DO 18 I=1,3
      ISIGN(I)=0
      XTMAX(I)=0.
18    CONTINUE
      DO 20 I=1,IJK
      HBAR(I)=0.
      HBARN(I)=0.
20    CONTINUE
C
C-- PROMPT FOR THE DATA FILE ON THE RIGHT-SIDE DISK
      CALL INPUT
C
C-- READ THE DATA FROM BOTH DISKS, FIND THE MAXIMUM PEAK,
C-- THE MINIMUM VALLEY, AND THE NUMBER POINTS PRESENT IN BOTH
C-- FILES.
      CALL MXMN
C
C-- DESIGNATE THE INTERVAL TO BE USED FOR PRINTING
C-- OUT THE HISTOGRAM OF THE STRESS RANGES.
      CALL HSTNT2
C
      REWIND INFILE
      REWIND 60
C-- THE LOAD-TIME HISTORY IS READ FROM THE TWO DISKS.  THE READ
C-- STARTS ON THE RIGHT-SIDE DISK AND THEN THE LEFT-SIDE DISK
C-- IS READ.
40    CONTINUE
      IF (IREAD .EQ. 1) GO TO 60
      READ(INFILE,50,END=500) IDAY,IHOUR,IMIN,SEC,XT
50    FORMAT(3I5,2E14.5)
      GO TO 100
60    READ(60,50,END=600) IDAY,IHOUR,IMIN,SEC,XT
100   TYPE 115, XT
```

```
 115   FORMAT(' THE VALUE OF XT: ',E14.5)
C
C-- TESTS TO DETERMINE IF THE RANGE IS TO BE COUNTED
C-- AND USED FOR FURTHER CALCULATIONS. THE PEAK AND
C-- VALLEY ARE TO BE THE ONLY POINTS TO BE CONSIDERED.
C-- THE TESTS EXAMINE SUCCESSIVE POINTS.  IF THEY ARE
C-- OF THE SAME SIGN, THEN THERE WAS NO ZERO-CROSSING,
C-- THUS, THE SMALLER OF THE TWO POINTS IS DISREGARDED
C-- FOR ANY FURTHER CALCULATIONS.  ONCE THERE HAVE BEEN
C-- THREE ZERO-CROSSINGS, A CYCLE AND STRESS RANGE ARE
C-- COUNTED.  THE PEAK AND VALLEY WHICH MAKE UP THIS RANGE
C-- ARE DISCARDED SINCE THEY HAVE NO EFFECT ON FUTURE
C-- EVENTS.
         IF (XT .LT. 0.) ISIGN(N)=0
         IF (XT .GE. 0.) ISIGN(N)=1
         XTABS=ABS(XT)
         IF (N .GT. 1) GO TO 150
         XTMAX(N)=XTABS
         N=N+1
         GO TO 40
 150     IF (ISIGN(N) .NE. ISIGN(N-1)) GO TO 200
         XTMAX(N-1)=AMAX1(XTMAX(N),XTABS)
         GO TO 40
 200     XTMAX(N)=XTABS
         IF (N .EQ. 3) GO TO 250
         N=N+1
         GO TO 40
 250     RANGE=XTMAX(1)+XTMAX(2)
C
C-- THE STRESS RANGES ARE PLACED INTO A HISTOGRAM
         CALL HISTGM(RANGE)
C
C-- THE COUNTED POINTS ARE DISCARDED AND THE COUNT CONTINUES.
         XTMAX(1)=XTMAX(2)
         XTMAX(2)=XTMAX(3)
         ISIGN(2)=ISIGN(3)
         N=3
         CYCLE=CYCLE+1.
C
C-- THE VALUES OF HBAR ARE CALCULATED USING THE STRESS RANGES
C-- OBTAINED FROM THE MEAN CROSSING-RANGE METHOD.
         DO 300 I=1,IJK
         HBAR(I)=(RANGE**YXZ)+HBAR(I)
         YXZ=YXZ+YHAT
         IF (I .EQ. IJK) YXZ=YZ
 300     CONTINUE
         GO TO 40
 500     IREAD=1
C
C-- THE RESULTS ARE NOW PRINTED OUT
         TYPE 510
 510     FORMAT(/,' END OF FILE IN DISK DK:')
         GO TO 40
 600     PRINT 610, IDAY,IHOUR,IMIN,SEC,XT
```

```
610   FORMAT(/,' RESULTS OF HBARN CYCLE COUNTING METHOD',///,
     1' THE TIME AND AMPLITUDE OF THE LAST DATA POINT ARE:',/,
     1I5,':',I2,':',I2,':',F6.3,';',E14.5,///)
C
C-- PRINT OUT THE RESULTS OF THE HISTOGRAM
      PRINT 900
900   FORMAT('    ----------- RANGE -----------',10X,
     1 '    -- CYCLE --')
      DO 930, I=1,NINC
      YRANGE(I)=(YRANGE(I)/2)
910   PRINT 920,XRANGE(I),XRANGE(I+1),YRANGE(I)
920   FORMAT(E14.5,'  ---',E13.5,10X,E14.5)
930   CONTINUE
C
C-- CALCULATE THE VALUE OF HBARN (H-ROOT) FOR THE OBTAINED
C-- STRESS RANGES USING THE RESPECTIVE EXPONENT, THEN THE
C-- RESULTS ARE PRINTED OUT.
      CYCL2=(CYCLE/2)
      PRINT 700,CYCL2
700   FORMAT(/,' THE TOTAL NUMBER OF CYCLES IS: ',E14.5)
      DO 800, I=1,IJK
      HBARN(I)=(HBAR(I)/CYCLE)**(1./YZ)
      PRINT 750,YZ,HBARN(I)
750   FORMAT(//,' WITH THE EXPONENT OF: ',E14.5,/,
     1  ' THE H-BAR OF THE RANGES IS: ',E14.5)
      YZ=YZ+YHAT
800   CONTINUE
      CALL EXIT
      END
```

```
      SUBROUTINE HISTGM(RANGE)
C
C-- PURPOSE:
C--   CALCULATES THE SIZE OF THE INTERVAL FOR THE HISTOGRAM
C--   PLACEMENT OF RANGES, PLACES THE OBTAINED STRESS RANGES
C--   INTO THE APPROPRIATE INTERVAL, AND THEN COUNTS THE
C--   NUMBER OF TIMES THE RANGES FALL WITHIN A GIVEN
C--   INTERVAL. NOTE, THIS IS HISTOGRAM DATA, AND NOT A
C--   HISTGRAM.
C-- USAGE:
C--   CALL HISTGM(RANGE)
C-- DESCRIPTION OF PARAMETERS:
C--   XRANGE - ARRAY OF THE ENDPOINTS OF THE INTERVAL
C--            USED FOR THE HISTOGRAM PLACEMENT
C--   RANGE  - STRESS RANGE OBTAINED FROM RAINFLOW
C--   XDEL1  - CHECKS TO SEE IF THE RANGE IS ABOVE THE
C--            LOWER ENDPOINT OF THE HISTOGRAM INTERVAL
C--   XDEL2  - CHECKS TO SEE IF THE RANGE IS BELOW THE
C--            UPPER ENDPOINT OF THE HISTOGRAM INTERVAL
C--   YRANGE - ARRAY TO COUNT THE NUMBER OF TIMES A RANGE
C--            MAGNITUDE FALLS WITHIN A GIVEN INTERVAL
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   NONE.
C-- COMMENTS:
C--   DOUBLE PRECISION IS USED TO MAKE SURE THAT ALL THE
C--   RANGES GET COUNTED AND PLACED INTO THE HISTOGRAM.
C--   MAINLY USED TO INCREASE THE ACCURACY OF THE
C--   COMPARISON/PLACEMENT TEST.
C
      IMPLICIT DOUBLE PRECISION (A,H,R,X)
      COMMON/HSTNT2/NINC,XINC,YRANGE(100),XRANGE(100)
C
C-- INITIALIZE THE PARAMETER
      XRANGE(1)=0.
C
C-- INCREMENT STEP TO ESTABLISH INTERVALS
      DO 10, I=2,NINC+1
      XRANGE(I)=XRANGE(I-1)+XINC
C
C-- * COMPARISON/PLACEMENT TEST *
C-- COMPARISON OF THE RANGES FOR HISTOGRAM PLACEMENT
      XDEL1=RANGE-XRANGE(I-1)
      XDEL2=RANGE-XRANGE(I)
      IF (XDEL1 .GT. 1.D-5 .AND. XDEL2 .LE. 1.D-5)
C
C-- COUNTS THE NUMBER OF OCCURANCES WITHIN AN INTERVAL
     1 YRANGE(I-1)=YRANGE(I-1)+1.
 10   CONTINUE
      RETURN
      END
```

```
      SUBROUTINE HSTNT2
C
C-- PURPOSE:
C--   PROMPTS FOR THE NUMBER OF INTERVALS TO BE USED IN
C--   SETTING UP A HISTGRAM FOR THE STRESS RANGES OBTAINED
C--   FROM ZERO CROSSING/MEAN.  ONCE THE NUMBER OF INTERVALS IS
C--   KNOWN, THE INCREMENT OF THE HISTOGRAM CAN BE ESTABLISHED.
C-- USAGE:
C--   CALL HSTNT2
C-- DESCRIPTION OF PARAMETERS:
C--   NINC   - THE TOTAL NUMBER OF INTERVALS BETWEEN
C--            EXTREMAS FOR THE HISTOGRAM
C--   RNGMAX - THE MAXIMUM POSSIBLE RANGE FOR THE HISTOGRAM
C--   XNINC  - THE REAL NUMBER CONVERSION OF NINC
C--   XINC   - THE INCREMENT FOR THE HISTOGRAM
C--   YRANGE - ARRAY TO COUNT THE NUMBER OF TIMES A RANGE
C--            MAGNITUDE FALLS WITHIN A GIVEN INTERVAL
C--   XRANGE - ARRAY OF THE ENDPOINTS OF THE INTERVALS
C--            USED FOR THE HISTOGRAM PLACEMENT
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   YESNO(IYESNO)
C-- COMMENTS:
C--   DOUBLE PRECISION IS USED TO INCREASE THE ACCURACY
C--   OF THE CALCULATIONS.
C
      IMPLICIT DOUBLE PRECISION (A,H,R,X)
      COMMON/MXMN/APEX,AVALLY,XPOINT
      COMMON/HSTNT2/NINC,XINC,YRANGE(100),XRANGE(100)
C
C-- THE MAXIMUM POSSIBLE RANGE FOR THE HISTOGRAM IS FOUND
      RNGMAX=APEX-AVALLY
 20   TYPE 30
 30   FORMAT(/,' TYPE IN THE NUMBER OF DESIRED INTERVALS',/,
     1' BETWEEN THE EXTREMA',I4,/)
      ACCEPT 40,NINC
 40   FORMAT(I3)
      TYPE 50,NINC
 50   FORMAT(/,' THE NUMBER OF INTERVALS IS: ',I5,/,
     1' IS THIS INFORMATION CORRECT?')
      CALL YESNO(IANS)
      IF (IANS .EQ. 0) GO TO 20
C
C-- REAL NUMBER CONVERSION
      XNINC=FLOAT(NINC)
      XINC=RNGMAX/XNINC
C
C-- INITIALIZE YRANGE FOR THE HISTOGRAM COUNT
      DO 60,I=1,NINC
      YRANGE(I)=0.
 60   CONTINUE
      RETURN
      END
```

```
      SUBROUTINE INPUT
C
C-- PURPOSE:
C--   DESIGNATE WHICH INPUT FILE WILL BE USED IN THE
C--   CALCULATIONS.  THIS DATA FILE SHOULD BE FOUND
C--   ON THE RIGHT-SIDE DISK.
C-- USAGE:
C--   CALL INPUT
C-- DESCRIPTION OF PARAMETERS:
C--   INFILE - THE NUMBER OF THE FILE TO BE EXAMINED
C--   IANS   - YES=1, OR NO=0
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   YESNO(IYESNO)
C-- COMMENTS:
C--   NONE.
C
      COMMON/INPUT/INFILE
20    TYPE 30
C
C-- DESIGNATE INPUT FILE TO BE USED
30    FORMAT(/,' TYPE IN THE NO. OF INPUT FILE',/)
      ACCEPT 40, INFILE
40    FORMAT(I3)
      TYPE 50, INFILE
50    FORMAT(//,'-- THERE WILL BE A SLIGHT WAIT',/,
     1 ' WHILE THIS FILE IS READ. --',//,
     1 ' THE INPUT FILE IS',I4,//,
     1' IS THIS INFORMATION CORRECT?')
      CALL YESNO(IANS)
      IF (IANS .EQ. 0) GO TO 20
      RETURN
      END
```

```
      SUBROUTINE MXMN
C
C-- PURPOSE:
C--   TO READ DATA FROM BOTH LEFT AND RIGHT DISKS, FIND THE
C--   MAXIMUM PEAK,THE MINIMUM VALLEY, AND THEN COUNT THE TOTAL
C--   NUMBER OF DATA POINTS.
C-- USAGE:
C--   CALL MXMN
C-- DESCRIPTION OF PARAMETERS:
C--   IREAD  - DETERMINES WHICH DISK IS TO BE READ
C--          0   - INFILE
C--          <>0 - VAXT.DAT (FTN60.DAT)
C--   XPOINT - THE TOTAL NUMBER OF DATA POINTS
C--   APEX   - MAXIMUM PEAK
C--   AVALLY - MINIMUM VALLEY
C--   IDAY   - NUMBER DAYS OF RANDOM LOAD
C--   IHOUR  - NUMBER HOURS OF RANDOM LOAD
C--   IMIN   - NUMBER OF MINUTES OF RANDOM LOAD
C--   XSEC   - NUMBER OF SECONDS OF RANDOM LOAD
C--   XH     - LOAD CORRESPONDING TO ABOVE TIMES
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   NONE.
C-- COMMENTS:
C--   DOUBLE-PRECISION WAS USED TO INCREASE THE ACCURACY
C--    OF THE CALCULATIONS.
C
      IMPLICIT DOUBLE PRECISION (A,H,R,X)
      COMMON/INPUT/INFILE
      COMMON/MXMN/APEX,AVALLY,XPOINT
C
C-- INITIALIZE THE PARAMETERS
      IREAD=0
      XPOINT=0.
      APEX=-10000.0
      AVALLY=10000.0
C
C-- XPOINT IS USED AS A COUNTER FOR EACH DATA FILE
C-- AND IS INCREASED BY ONE EACH TIME A DATA POINT IS
C-- READ, RESULTING IN A TOTAL COUNT OF ALL DATA POINTS
C-- IN THE INPUT FILES.
10    XPOINT=XPOINT+1.
C
C-- THE LOAD-TIME HISTORY IS READ FROM EACH DATA FILE.
C-- THE READING STARTS WITH THE RIGHT-SIDE DISK AND ONCE
C-- THAT DISK IS FINISHED, THE LEFT-SIDE DISK IS READ.
      IF (IREAD .NE. 0) GO TO 21
C
C-- ONCE THE END OF THE FILE IS REACHED, IREAD MUST BE MADE
C-- GREATER THAN ZERO TO INDICATE THE LEFT DISK SHOULD BE READ.
      READ(INFILE,20,END=30) IDAY,IHOUR,IMIN,XSEC,XH
20    FORMAT(3I5,2F14.5)
      GO TO 25
C
C-- ONCE THE LEFT-SIDE DISK HAS BEEN COMPLETELY READ,
```

```
C-- THE RESULTS CAN BE PRINTED OUT IF SO DESIRED.
 21     READ(60,20,END=40) IDAY,IHOUR,IMIN,XSEC,XH
C
C-- THE COMPARISON TESTS TO FIND THE PEAK AND VALLEY OF FILES
 25     IF (XH .GT. APEX) APEX=XH
        IF (XH .LT. AVALLY) AVALLY=XH
        GO TO 10
 30     IREAD=IREAD+1
        GO TO 21
C
C-- PRINTS THE PEAK AND VALLEY
 40     PRINT 41,APEX,AVALLY
 41     FORMAT(//' THE PEAK IS ',D14.5,/,' THE VALLEY IS ',D14.5)
        XPOINT=XPOINT-1.
C
C-- PRINTS THE NUMBER OF DATA POINTS PRESENT IN BOTH FILES
        PRINT 60,XPOINT
 60     FORMAT(/' THERE ARE ',D14.5,' POINTS.'//)
        RETURN
        END
```

```
      SUBROUTINE TEXPO
C
C-- PURPOSE:
C--   DESIGNATES THE STARTING EXPONENT, THE NUMBER OF H-ROOT
C--   CALCULATIONS, AND THE INCREMENT BY WHICH THE EXPONENT
C--   IS INCREASED FOR EACH H-ROOT CALCULATION.
C-- USAGE:
C--   CALL TEXPO
C-- DESCRIPTION OF PARAMETERS:
C--   YZ     - STARTING EXPONENT FOR THE H-ROOT CALCULATIONS
C--   IJK    - THE NUMBER OF TIMES H-ROOT IS TO BE CALCULATED
C--   YHAT   - THE INCREMENT BY WHICH THE EXPONENT IS INCREASED
C--             FOR EACH H-ROOT CALCULATION
C-- SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED:
C--   YESNO(IYESNO)
C-- COMMENTS:
C--   NONE.
C
      COMMON/EXPO/YZ,IJK,YHAT
20    TYPE 30
30    FORMAT(/,' TYPE STARTING EXPONENT FOR HROOT')
      ACCEPT 40,YZ
40    FORMAT(F7.4)
      TYPE 42
42    FORMAT(/,' TYPE THE NUMBER OF H-ROOT CALCULATIONS')
      ACCEPT 45,IJK
45    FORMAT(I5)
      TYPE 46
46    FORMAT(/,' TYPE THE INCREMENT FOR H--ROOT EXPONENT')
      ACCEPT 47,YHAT
47    FORMAT(F7.5)
      TYPE 50,YZ
50    FORMAT(/,' THE STARTING EXPONENT OF HROOT IS: ',F7.4,/)
      TYPE 55,IJK
55    FORMAT(' THE NUMBER OF H-ROOT CALCULATIONS IS: ',I5,/)
      TYPE 60,YHAT
60    FORMAT(' THE INCREMENT OF H-ROOT EXPONENT IS: ',F7.5,//,
     1 ' IS THIS INFORMATION CORRECT?')
      CALL YESNO(IANS)
      IF (IANS .EQ. 0) GO TO 20
      RETURN
      END
```

```
      SUBROUTINE YESNO(IYESNO)
C
C-- THIS IS A SIMPLE DOUBLE CHECK TEST PROGRAM
C-- CAN BE USED ANYWHERE AN INPUT IS TO BE
C-- DOUBLE CHECKED.
      LOGICAL*1 ANS(20)
      LOGICAL*1 CHARN
      DATA CHARN/1HN/
      TYPE 2
2     FORMAT(' ANSWER YES OR NO')
      ACCEPT 3, ANS
3     FORMAT(20A1)
      DO 70 I=1,20
70    IF(ANS(I).EQ.CHARN)GOTO  80
      IYESNO=1
      RETURN
80    IYESNO=0
      END
```

# APPENDIX 3

## EXAMPLE OF PROGRAM EXECUTION

## (APPLICABLE FOR BOTH PROGRAMS IN APPENDIXES 1 AND 2)

## INPUT DATA

For the programs to function properly, two input data sets are required:

1. Data File: VAXT.DAT (FTN 60.DAT)

   This file, which contains the second portion of the data to be analyzed must be on the system volume, the "left-disk," for the programs to function properly. On the system volume, the file is called VAXT.DAT; however, in the programs it will be assigned the name FTN 60.DAT. Refer to figure A3.1 for sample file.

2. Data File: FTN xx.DAT

   This file, which contains the first portion of the data to be analyzed, is on the storage volume, the "right-disk." Refer to figure A3.2 for sample file. The symbol xx represents an integer whose value is between 1 and 99.

NOTE: FTN xx.DAT is read first and VAXT.DAT (FTN 60.DAT) is read second, but this program examines these two files as if they were one large data file.

## RUNNING OF THE PROGRAMS

The programs were written in the interactive mode, refer to figure A3.3 for an example of the prompts.

Description of Terms:

Intervals between extrema - the number of divisions between o and the largest range when generating the histogram data for the stress ranges.

HROOT - equivalent-stress range, defined to be:

| DAY | HR | MIN | SEC | VALUE OF PEAK OR VALLEY |
|---|---|---|---|---|
| 0 | 8 | 0 | .50078E+01 | .60965E+00 |
| 0 | 8 | 0 | .81312E+01 | -.85032E+00 |
| 0 | 8 | 0 | .11236E+02 | .79637E+00 |
| 0 | 8 | 0 | .14151E+02 | -.46743E+00 |
| 0 | 8 | 0 | .16519E+02 | .17717E+00 |
| 0 | 8 | 0 | .18854E+02 | -.46511E+00 |
| 0 | 8 | 0 | .21751E+02 | .98606E+00 |
| 0 | 8 | 0 | .24864E+02 | -.12579E+01 |
| 0 | 8 | 0 | .28020E+02 | .11645E+01 |
| 0 | 8 | 0 | .31139E+02 | -.83102E+00 |
| 0 | 8 | 0 | .34133E+02 | .46618E+00 |
| 0 | 8 | 0 | .36989E+02 | -.31618E+00 |
| 0 | 8 | 0 | .40084E+02 | .38693E+00 |
| 0 | 8 | 0 | .43541E+02 | -.58993E+00 |
| 0 | 8 | 0 | .46893E+02 | .85384E+00 |
| 0 | 8 | 0 | .50105E+02 | -.10866E+01 |
| 0 | 8 | 0 | .53259E+02 | .12281E+01 |
| 0 | 8 | 0 | .56420E+02 | -.12587E+01 |
| 0 | 8 | 0 | .59655E+02 | .11573E+01 |
| 0 | 8 | 1 | .30924E+01 | -.91464E+00 |
| 0 | 8 | 1 | .70484E+01 | .72587E+00 |
| 0 | 8 | 1 | .10831E+02 | -.75006E+00 |
| 0 | 8 | 1 | .14120E+02 | .69061E+00 |
| 0 | 8 | 1 | .17171E+02 | -.48633E+00 |
| 0 | 8 | 1 | .19799E+02 | .30465E+00 |
| 0 | 8 | 1 | .22784E+02 | -.31261E+00 |
| 0 | 8 | 1 | .24936E+02 | .47692E+00 |
| 0 | 8 | 1 | .27794E+02 | -.65215E+00 |
| 0 | 8 | 1 | .30775E+02 | .78700E+00 |
| 0 | 8 | 1 | .33859E+02 | -.84711E+00 |

Figure A3.1. Example of VAXT.DAT (FTN60.DAT) input file.

| DAY | HR | MIN | SEC | VALUE OF PEAK OR VALLEY |
|---|---|---|---|---|
| 0 | 0 | 0 | .14625E+01 | -.11702E+01 |
| 0 | 0 | 0 | .47861E+01 | .85788E+00 |
| 0 | 0 | 0 | .83945E+01 | -.41536E+00 |
| 0 | 0 | 0 | .13696E+02 | .45695E+00 |
| 0 | 0 | 0 | .16945E+02 | -.56073E+00 |
| 0 | 0 | 0 | .19937E+02 | .52202E+00 |
| 0 | 0 | 0 | .22754E+02 | -.49976E+00 |
| 0 | 0 | 0 | .25576E+02 | .63183E+00 |
| 0 | 0 | 0 | .28564E+02 | -.85585E+00 |
| 0 | 0 | 0 | .31706E+02 | .10121E+01 |
| 0 | 0 | 0 | .34937E+02 | -.10251E+01 |
| 0 | 0 | 0 | .38188E+02 | .91595E+00 |
| 0 | 0 | 0 | .41372E+02 | -.78927E+00 |
| 0 | 0 | 0 | .44447E+02 | .76372E+00 |
| 0 | 0 | 0 | .47486E+02 | -.91998E+00 |
| 0 | 0 | 0 | .50558E+02 | .11576E+01 |
| 0 | 0 | 0 | .53635E+02 | -.12860E+01 |
| 0 | 0 | 0 | .56652E+02 | .12600E+01 |
| 0 | 0 | 0 | .59562E+02 | -.11635E+01 |
| 0 | 0 | 1 | .23969E+01 | .11802E+01 |
| 0 | 0 | 1 | .52780E+01 | -.13601E+01 |
| 0 | 0 | 1 | .82657E+01 | .15178E+01 |
| 0 | 0 | 1 | .11324E+02 | -.14430E+01 |
| 0 | 0 | 1 | .14376E+02 | .10948E+01 |
| 0 | 0 | 1 | .17316E+02 | -.65628E+00 |
| 0 | 0 | 1 | .20048E+02 | .40073E+00 |
| 0 | 0 | 1 | .22779E+02 | -.41940E+00 |
| 0 | 0 | 1 | .25824E+02 | .45603E+00 |
| 0 | 0 | 1 | .29209E+02 | -.39155E+00 |
| 0 | 0 | 1 | .32737E+02 | .27893E+00 |

Figure A3.2. Example of FTNxx.DAT (FTN50.DAT) input file.

TYPE IN THE NO. OF INPUT FILE.

*50*

**THERE WILL BE A SLIGHT WAIT
WHILE THIS FILE IS READ.

THE INPUT FILE IS 50
IS THIS INFORMATION CORRECT?
ANSWER YES OR NO

*yes*

TYPE IN THE NUMBER OF DESIRED INTERVALS
BETWEEN THE EXTREMAS.

*20*

THE NUMBER OF INTERVALS IS : 20
IS THIS INFORMATION CORRECT?
ANSWER YES OR NO

*yes*

TYPE IN THE STARTING EXPONENT FOR HROOT

*2.0*

TYPE IN THE NUMBER FOR HROOT CALCULATIONS

*7*

TYPE IN THE INCREMENT FOR HROOT EXPONENT

*0.5*

THE STARTING EXPONENT FOR HROOT IS: 2.0000

THE NUMBER FOR HROOT CALCULATIONS IS: 7

THE INCREMENT FOR HROOT EXPONENT IS: 0.50000

IS THIS INFORMATION CORRECT?
ANSWER YES OR NO

*yes*

Figure A3.3. Example of program execution. Italic letters are the operator's responses to the computer's prompts.

$$\text{HROOT} = \sqrt[n]{\frac{\sum_{i=1}^{N} H_i^n}{N}}$$

where,

$n$ = some power (exponent in fatigue crack growth equation)

$H_i$ = ith stress range

$N$ = total number of ranges

OUTPUT

The output is directly printed out and consists of the extrema, the number of data points analyzed, a histogram data of the calculated stress ranges, the total number of stress ranges, and value of HROOT for its respective exponent. Refer to figures A3.4 and A3.5 for examples of output for each program (rainflow, and mean crossing-range techniques, respectively).

THE PEAK IS 0.15178D+01
THE VALLEY IS -0.14430D+01

THERE ARE 0.60000D+02 POINTS.

| ----------- RANGE ----------- | | | -- CYCLE -- |
|---|---|---|---|
| 0.00000D+00 | --- | 0.14804D+00 | 0.00000E+00 |
| 0.14804D+00 | --- | 0.29608D+00 | 0.00000E+00 |
| 0.29608D+00 | --- | 0.44412D+00 | 0.00000E+00 |
| 0.44412D+00 | --- | 0.59216D+00 | 0.00000E+00 |
| 0.59216D+00 | --- | 0.74020D+00 | 0.30000E+01 |
| 0.74020D+00 | --- | 0.88824D+00 | 0.30000E+01 |
| 0.88824D+00 | --- | 0.10363D+01 | 0.20000E+01 |
| 0.10363D+01 | --- | 0.11843D+01 | 0.10000E+01 |
| 0.11843D+01 | --- | 0.13324D+01 | 0.30000E+01 |
| 0.13324D+01 | --- | 0.14804D+01 | 0.20000E+01 |
| 0.14804D+01 | --- | 0.16284D+01 | 0.10000E+01 |
| 0.16284D+01 | --- | 0.17765D+01 | 0.30000E+01 |
| 0.17765D+01 | --- | 0.19245D+01 | 0.20000E+01 |
| 0.19245D+01 | --- | 0.20726D+01 | 0.10000E+01 |
| 0.20726D+01 | --- | 0.22206D+01 | 0.00000E+00 |
| 0.22206D+01 | --- | 0.23686D+01 | 0.40000E+01 |
| 0.23686D+01 | --- | 0.25167D+01 | 0.20000E+01 |
| 0.25167D+01 | --- | 0.26647D+01 | 0.10000E+01 |
| 0.26647D+01 | --- | 0.28128D+01 | 0.00000E+00 |
| 0.28128D+01 | --- | 0.29608D+01 | 0.10000E+01 |

THE NUMBER OF RANGES IS: 0.29000D+02

WITH THE EXPONENT OF: 0.20000E+01
THE H-ROOT OF THE RANGES IS: 0.17762D+01

WITH THE EXPONENT OF: 0.25000E+01
THE H-ROOT OF THE RANGES IS: 0.17850D+01

WITH THE EXPONENT OF: 0.30000E+01
THE H-ROOT OF THE RANGES IS: 0.18390D+01

WITH THE EXPONENT OF: 0.35000E+01
THE H-ROOT OF THE RANGES IS: 0.18883D+01

WITH THE EXPONENT OF: 0.40000E+01
THE H-ROOT OF THE RANGES IS: 0.19333D+01

WITH THE EXPONENT OF: 0.45000E+01
THE H-ROOT OF THE RANGES IS: 0.19744D+01

WITH THE EXPONENT OF: 0.50000E+01
THE H-ROOT OF THE RANGES IS: 0.20121D+01

Figure A3.4. Output example from the rainflow cycle-counting method.

THERE ARE     0.60000D+02 POINTS.

| ----------- RANGE ----------- | | | -- CYCLE -- |
|---|---|---|---|
| 0.00000E+00 | --- | 0.14804E+00 | 0.00000E+00 |
| 0.14804E+00 | --- | 0.29608E+00 | 0.00000E+00 |
| 0.29608E+00 | --- | 0.44412E+00 | 0.00000E+00 |
| 0.44412E+00 | --- | 0.59216E+00 | 0.00000E+00 |
| 0.59216E+00 | --- | 0.74020E+00 | 0.20000E+01 |
| 0.74020E+00 | --- | 0.88824E+00 | 0.35000E+01 |
| 0.88824E+00 | --- | 0.10363E+01 | 0.20000E+01 |
| 0.10363E+01 | --- | 0.11843E+01 | 0.25000E+01 |
| 0.11843E+01 | --- | 0.13324E+01 | 0.15000E+01 |
| 0.13324E+01 | --- | 0.14804E+01 | 0.30000E+01 |
| 0.14804E+01 | --- | 0.16284E+01 | 0.10000E+01 |
| 0.16284E+01 | --- | 0.17765E+01 | 0.25000E+01 |
| 0.17765E+01 | --- | 0.19245E+01 | 0.50000E+00 |
| 0.19245E+01 | --- | 0.20726E+01 | 0.30000E+01 |
| 0.20726E+01 | --- | 0.22206E+01 | 0.50000E+00 |
| 0.22206E+01 | --- | 0.23686E+01 | 0.15000E+01 |
| 0.23686E+01 | --- | 0.25167E+01 | 0.25000E+01 |
| 0.25167E+01 | --- | 0.26647E+01 | 0.15000E+01 |
| 0.26647E+01 | --- | 0.28128E+01 | 0.00000E+00 |
| 0.28128E+01 | --- | 0.29608E+01 | 0.10000E+01 |

THE TOTAL NUMBER OF CYCLES IS:     0.28500E+02

WITH THE EXPONENT OF:     0.20000E+01
THE H-BAR OF THE RANGES IS:     0.17120E+01

WITH THE EXPONENT OF:     0.25000E+01
THE H-BAR OF THE RANGES IS:     0.17681E+01

WITH THE EXPONENT OF:     0.30000E+01
THE H-BAR OF THE RANGES IS:     0.18200E+01

WITH THE EXPONENT OF:     0.35000E+01
THE H-BAR OF THE RANGES IS:     0.18677E+01

WITH THE EXPONENT OF:     0.40000E+01
THE H-BAR OF THE RANGES IS:     0.19116E+01

WITH THE EXPONENT OF:     0.45000E+01
THE H-BAR OF THE RANGES IS:     0.19519E+01

WITH THE EXPONENT OF:     0.50000E+01
THE H-BAR OF THE RANGES IS:     0.19889E+01

Figure A3.5. Output example from the mean crossing-range cycle-counting technique.

NBS-114A (REV. 2-80)

| U.S. DEPT. OF COMM. BIBLIOGRAPHIC DATA SHEET (See instructions) | 1. PUBLICATION OR REPORT NO. NBSIR 86-3055 | 2. Performing Organ. Report No. | 3. Publication Date August 1986 |
|---|---|---|---|

**4. TITLE AND SUBTITLE**

CYCLE-COUNTING METHODS FOR FATIGUE ANALYSIS WITH RANDOM LOAD HISTORIES: A FORTRAN USER'S GUIDE

**5. AUTHOR(S)**

Yi-Wen Cheng and Jerry J. Broz

**6. PERFORMING ORGANIZATION** *(If joint or other than NBS, see instructions)*

NATIONAL BUREAU OF STANDARDS
DEPARTMENT OF COMMERCE
WASHINGTON, D.C. 20234

**7. Contract/Grant No.**

**8. Type of Report & Period Covered**

**9. SPONSORING ORGANIZATION NAME AND COMPLETE ADDRESS** *(Street, City, State, ZIP)*

U. S. Department of Interior
Minerals Management Service
12203 Sunrise Valley Drive
Reston, Virginia 22091

**10. SUPPLEMENTARY NOTES**

☐ Document describes a computer program; SF-185, FIPS Software Summary, is attached.

**11. ABSTRACT** *(A 200-word or less factual summary of most significant information. If document includes a significant bibliography or literature survey, mention it here)*

Rainflow and mean crossing-range methods are used in counting the stress ranges and cycles of a random load history. Each method is defined and then applied to a simple random load history example. Fortran IV computer programs were written to make analysis of long random load histories possible. The stress ranges and cycles obtained by these programs have been used for fatigue crack growth analysis under sea-wave loading.

**12. KEY WORDS** *(Six to twelve entries; alphabetical order; capitalize only proper names; and separate key words by semicolons)*

cycle-counting methods; fatigue of materials; mean crossing-range counting method; rainflow counting method; variable amplitude loading.

**13. AVAILABILITY**

☒ Unlimited
☐ For Official Distribution. Do Not Release to NTIS
☐ Order From Superintendent of Documents, U.S. Government Printing Office, Washington, D.C. 20402.
☒ Order From National Technical Information Service (NTIS), Springfield, VA. 22161

**14. NO. OF PRINTED PAGES**

52

**15. Price**